
Model de creștere economică Ramsey

Neculai Andrei,

*Institutul de Cercetări în Informatică,
Centrul de Modelare și Optimizare Avansată
8-10, Bdl. Averescu, București 1, România,
nandrei@ici.ro*

Rezumat. Prezentăm modelul de creștere economică Ramsey, precum și reprezentarea și soluția acestuia în limbajul GAMS.

1. Modelul Ramsey

Teoria creșterii economice se bazează pe lucrările de pionierat ale lui Frank Ramsey [1928], care în scurt timp au devenit un instrument standard în modelarea economică aplicată. În domeniul economic Ramsey a publicat două lucrări: *A contribution to the theory of taxation* și *A mathematical theory of saving*. Acestea au avut un impact deosebit asupra teoriei matematice a economiei, determinând utilizarea modelelor din ce în ce mai sofisticate în acest domeniu.

Pentru a defini un model de creștere economică să introducem: C_t consumul la momentul de timp t and ρ rata de discount. Fie în același timp $u(C_t)$ o funcție de utilitate instantanee care se bucură de următoarele proprietăți: $u(C_t) \geq 0$, $u'(C_t) > 0$ și $u''(C_t) < 0$. Cu acestea, maximizarea utilității (preferința consumatorului) conduce la următoarea funcție de bunăstare:

$$\max W = \int_{t=0}^{\infty} u(C_t) e^{-\rho t} dt.$$



Frank Ramsey (1903-1930)

În continuare să introducem $f(K_t)$ funcția de producție dependentă de capitalul K_t , cu următoarele proprietăți: $f(0) = 0$, $f'(0) = \infty$ și $f'(\infty) = 0$, (Inada conditions [1963]).

Cu acestea ecuația de acumulare a capitalului se definește prin intermediul restricției bugetare sub forma:

$$f(K_t) = C_t + \frac{dK}{dt},$$

unde termenul dK / dt reprezintă investiția netă. Cu alte cuvinte, ieșirea este consumată sau investită.

Studiul analitic al acestui model este prezentat în [Barro și Martin, 1998], [Barro, 1998, 1999], [Pedersen, 1999]. Pentru studiul comportării numerice vom introduce o variantă discretă a modelului de mai sus sub forma [Ramsey, 1928], [Kalvelagen, 2003]:

$$\max W = \sum_{t=0}^{\infty} \left(\frac{1}{1 + \rho} \right)^t u(C_t)$$

referitor la:

$$Y_t = f(K_t)$$

$$Y_t = C_t + I_t$$

$$K_{t+1} = (1 - \delta)K_t + I_t,$$

unde δ este rata de depreciere a capitalului.

De obicei, funcția de utilitate $u(\cdot)$ și funcția de producție $f(\cdot)$ se definesc sub forma:

$$u(C_t) = \log(C_t),$$

$$f(K_t) = aK_t^b L_t^{1-b},$$

care este o funcție de tip Cobb-Douglas, unde L_t este factorul de producție determinat sub forma:

$$L_{t+1} = (1 + g)L_t,$$

unde g este rata de creștere a producției.

Pentru a utiliza acest model este necesar să tratăm orizontul de timp infinit din funcția obiectiv. O soluție, foarte simplă și la îndemână este da a ignora termenii seriei de la un anumit moment de timp $t = T$. Totuși, o soluție mai rafinată consideră $C_t = C_T$ pentru $t > T$, adică de la un moment T încolo consumul este constant.

În acest moment să introducem o mică digresiune. După cum știm pentru orice $0 < a < 1$, putem scrie:

$$\sum_{t=0}^{\infty} a^t = \lim_{T \rightarrow \infty} \sum_{t=0}^{T-1} a^t = \lim_{T \rightarrow \infty} \frac{1 - a^T}{1 - a} = \frac{1}{1 - a}.$$

Dar,

$$\sum_{t=0}^{T-1} a^t = \frac{1 - a^T}{1 - a}$$

și deci,

$$\sum_{t=T}^{\infty} a^t = \sum_{t=0}^{\infty} a^t - \sum_{t=0}^{T-1} a^t = \frac{1}{1 - a} - \frac{1 - a^T}{1 - a} = \frac{a^T}{1 - a}.$$

Cu acestea, funcția obiectiv devine;

$$\begin{aligned} \max W &= \sum_{t=0}^{T-1} \left(\frac{1}{1+\rho} \right)^t \log(C_t) + \sum_{t=T}^{\infty} \left(\frac{1}{1+\rho} \right)^t \log(C_T) \\ &= \sum_{t=0}^{T-1} \left(\frac{1}{1+\rho} \right)^t \log(C_t) + \frac{1}{\rho(1+\rho)^{T-1}} \log(C_T) \\ &= \sum_{t=0}^T \beta_t \log(C_t), \end{aligned}$$

unde:

$$\beta_t = \begin{cases} (1+\rho)^{-t} & , \quad t < T, \\ \rho^{-1}(1+\rho)^{1-T} & , \quad t = T. \end{cases}$$

Observăm că astfel formulată, problema asociază o pondere valorii terminale $u(C_T)$. În experimentele de mai jos vom vedea că această pondere este importantă. Mai mult, acest mod de abordare a problemelor de optimizare cu orizont de timp infinit se poate aplica și în alte contexte.

Trunchierea funcției obiectiv impune introducerea unei restricții la momentul final T referitoare la cerința unei investiții minime cu care să se restarteze mecanismul economic:

$$I_T \geq (g + \delta)K_T,$$

unde, după cum am văzut, g este rata de creștere a producției și δ este rata de depreciere a capitalului.

Cu acestea un model de creștere economică de tip Ramsey se prezintă sub forma:

$$\max W = \sum_{t=0}^T \beta_t \log(C_t),$$

referitor la:

$$\begin{aligned} Y_t &= a K_t^b L_t^{1-b}, & t = 0, 1, \dots, T, \\ Y_t &= C_t + I_t, & t = 0, 1, \dots, T, \\ K_{t+1} &= (1-\delta)K_t + I_t, & t = 0, 1, \dots, T-1, \\ I_T &\geq (g + \delta)K_T, \end{aligned}$$

unde ρ (factorul de discaunt), g (rata de creștere a producției), δ (factorul de depreciere a capitalului), K_0 (capitalul inițial), I_0 (investiția inițială), C_0 (consumul inițial), L_0 (factorul de producție inițial), b și a (coeficienți Cobb-Douglas) sunt parametri cunoscuți.

Observăm imediat că coeficientul Cobb-Douglas a se poate determina sub forma următoare. Avem: $Y_0 = C_0 + I_0$ și $Y_0 = f(K_0, L_0) = a K_0^b L_0^{1-b}$, deci

$$a = \frac{C_0 + I_0}{K_0^b L_0^{1-b}}.$$

2. Expresia modelului în GAMS

```

$ontext
  Model de crestere economica (echilibru dinamic) Ramsey
  Bibliografie:
    Frank P. Ramsey, (1928) A Mathematical Theory of Saving,
    Economics Journal, December 1928.
    Erwin Kalvelagen, (2003) An elementary Ramsey growth
    model. http://www.gams.com/~erwin/micro/growth.gms
$offtext
*
* Date asociate modelului:
*
set t      'perioadele de timp'          / t1*t50 /

SCALARS
  rho      'factorul de discaunt'          / 0.04 /
  g        'rata de crestere a productiei' / 0.03 /
  delta    'factorul de depreciere al capitalului' / 0.02 /
  K0       'capitalul initial'            / 3.00 /
  I0       'investitia initiala'          / 0.07 /
  C0       'consumul initial'             / 0.95 /
  L0       'factorul de productie initial' / 1.00 /
  b        'coeficient Cobb Douglas'      / 0.25 /
  a        'coeficient Cobb Douglas'
;
*
* Multimi de lucru
*
SETS
  tfirst(t) 'primul interval (t0)'
  tlast(t)  'ultimul interval (T)'
  tnotlast(t) 'toate intervalele de timp fara ultimul'
;
tfirst(t)$ (ord(t)=1) = yes;
tlast(t)$ (ord(t)=card(t)) = yes;
tnotlast(t) = not tlast(t);
*
PARAMETERS
  L(t)      'factorul de productie'
  beta(t)   'ponderea utilitatilor in functia obiectiv'
  tval(t)   'valoarea numerica a lui t'
;
tval(t) = ord(t)-1;
*
* Calculul ponderilor utilitatilor.
* Pondere asociata ultimului interval de timp (ponderea
* terminala) compenseaza folosirea utilitatilor dincolo de

```

```

* momentul de timp T.
*
beta(tnotlast(t)) = power(1+rho,-tval(t));
beta(tlast(t)) = (1/rho)*power(1+rho,1-tval(t));
display beta;
*
* Calculul factorului de productie utilizand un proces
* de crestere exponentiala.
*
L(t) = power(1+g,tval(t)) * L0;
*
* Calculul coeficientului Cobb-Douglas a.
*
a = (C0 + I0) / (K0**b * L0**(1-b));
*
VARIABLES
      C(t)      'consumul'
      Y(t)      'productia'
      K(t)      'capitalul'
      I(t)      'investitia'
      W          'utilitatea totala'
;
*
EQUATION
      utilitate      'functia obiectiv, utilitatea'
      productie(t)  'functia de productie Cobb-Douglas'
      alocare(t)     'ecuatia productiei'
      acumulare(t)  'ecuatia acumularii capitalului'
      final(t)      'investitia in ultimul interval'
;
utilitate..      W =e= sum(t, beta(t)*log(C(t)));
productie(t)..  Y(t) =e= a * (K(t)**b) * (L(t)**(1-b));
alocare(t)..    Y(t) =e= C(t) + I(t);
acumulare(tnotlast(t)).. K(t+1) =e= (1-delta)*K(t) + I(t);
final(tlast)..  I(tlast) =g= (g+delta)*K(tlast);
*
* Margini asupra unor variabile pentru a putea
* evalua functiile.
*
K.lo(t) = 0.001;
C.lo(t) = 0.001;
*
* Condițiile initiale.
*
K.fx(tfirst) = K0;
I.fx(tfirst) = I0;
C.fx(tfirst) = C0;

model ramsey /all/;
solve ramsey maximizing W using nlp;
*
* Afisarea solutiei
*

```

```

file res1 /growth.txt/
put res1;
put /"time    C(t)  Y(t)  K(t)  I(t)  "/;
loop(t, put t.tl:6, C.l(t):6, Y.l(t):6, K.l(t):6, I.l(t):6
/;);
*End of model

```

Soluția modelului, corespunzătoare datelor precizate este prezentată în figura 1.




Fig.1. Soluția modelului Ramsey.

Expresia GAMS a modelului conține instrucțiunea de afișare a ponderilor β_t din funcția obiectiv (display beta;). Valorile acestor ponderi sunt următoarele:

```

---- 51 PARAMETER beta    ponderea utilitatilor in functia obiectiv
t1  1.000,    t2  0.962,    t3  0.925,    t4  0.889,    t5  0.855,    t6
0.822
t7  0.790,    t8  0.760,    t9  0.731,    t10 0.703,    t11 0.676,    t12
0.650
t13 0.625,    t14 0.601,    t15 0.577,    t16 0.555,    t17 0.534,    t18
0.513
t19 0.494,    t20 0.475,    t21 0.456,    t22 0.439,    t23 0.422,    t24
0.406
t25 0.390,    t26 0.375,    t27 0.361,    t28 0.347,    t29 0.333,    t30
0.321

```

t31 0.308, 0.253	t32 0.296,	t33 0.285,	t34 0.274,	t35 0.264,	t36
t37 0.244, 0.200	t38 0.234,	t39 0.225,	t40 0.217,	t41 0.208,	t42
t43 0.193, 0.158	t44 0.185,	t45 0.178,	t46 0.171,	t47 0.165,	t48
t49 0.152,	t50 3.805				

Observăm că ponderea asociată momentului final β_{50} are o valoare substanțială. Aceasta arată încă odată importanța tehnicii de transformare a seriei infinite din funcția obiectiv într-o sumă finită.

O variantă de model Ramsey cu 30 de variabile și 21 de restricții este prezentată în [Andrei, 2003, pag. 398], unde se prezintă și soluția acestuia obținută cu pachetul SPENBAR [Andrei, 1996a-e, 1999].

Modele dinamice de tip Ramsey sunt foarte des utilizate în contextul modelării economice. Putem menționa modelul *DICE - Dynamic Integrated Climate-Economy* dezvoltat de Nordhaus [1992]; modelul *TAVF - Transitional Alternative Fuels and Vehicles* prezentat de Leiby și Rubin [1997] și *MEEN - Modeling the Environment-Economy Nexus* al lui Harrison [2001]. O clasă specială de modele bazată pe teoria modelelor dinamice Ramsey este dată de modelele *MARKAL-MACRO* [Manne și Wene, 1992].

Bibliografie

- Andrei, N.**, (1996a) Computational experience with a modified penalty-barrier method for large-scale nonlinear constrained optimization. Working Paper No. AMOL-96-1, Research Institute for Informatics, ICI, Bucharest, February 6, 1996
- Andrei, N.**, (1996b) Computational experience with a modified penalty-barrier method for large-scale nonlinear, equality and inequality constrained optimization. Technical Paper No. AMOL-96-2, Research Institute for Informatics, ICI, Bucharest, February 12, 1996.
- Andrei, N.**, (1996c) Computational experience with "SPENBAR" a sparse variant of a modified penalty-barrier method for large-scale nonlinear, equality and inequality, constrained optimization. Working Paper No. AMOL-96-3, Research Institute for Informatics, ICI, Bucharest, March 10, 1996.
- Andrei, N.**, (1996d) Computational experience with "SPENBAR" a sparse variant of a modified penalty-barrier method for large-scale nonlinear, equality and inequality, constrained optimization. Technical Paper No. AMOL-96-4, Research Institute for Informatics, ICI, Bucharest, March 11, 1996.
- Andrei, N.**, (1996e) Numerical examples with "SPENBAR" for large-scale nonlinear, equality and inequality, constrained optimization with zero columns in Jacobian matrices. Technical Paper No. AMOL-96-5, Research Institute for Informatics, ICI, Bucharest, March 29, 1996.
- Andrei, N.**, (1999) *Programarea Matematică Avansată. Teorie, Metode Computaționale, Aplicații*. Editura Tehnică - București, 1999.
- Andrei, N.**, (2003) *Modele, Probleme de Test și Aplicații de Programare Matematică*. Editura Tehnică, București, 2003.
- Barro, R., Sala-i-Martin, X.**, (1998) *Economic growth*, MIT Press, 1998.

- Barro, R.**, (1998) *Notes on growth accounting*. Harvard University Technical Report, December 17, 1998.
- Barro, R.**, (1999) *Ramsey meets Laibson in the neoclassical growth model*. Harvard University Technical Report, September, 1999.
- Daene McKinney și Andre Savitsky**, (2003) *Basic optimization models for water and energy management*, The University of Texas at Austin Technical Report, 2003.
- Harrison, G.W.**, (2001) *MEEN, modeling the environment-economy nexus*. http://theweb.badm.sc.edu/glenn/meen_mod.htm, 2001.
- Ken-Ichi Inada**, (1963) *On a two-sector model of economic growth: comments and a generalization*. Review of Economic Studies 30, pp.119-127, 1963.
- Kalvelagen, E.**, (2003) *An elementary Ramsey growth model*. GAMS Technical Report, March 12, 2003.
- Leiby, P., și Rubin, J.**, (1997) *Technical documentation of the transitional alternative fuels and vehicles (TAFV) model*. Technical Report, Oak Ridge National Laboratory, July 1997.
- Manne, A., și Wene, C.**, (1992) *MARKAL-MACRO: a linked model for energy-economy analysis*. Technical Report BNL-47161, Brookhaven National Laboratory, 1992.
- Nordhaus, W.D.** (1992) *An optimal transition path for controlling greenhouse gasses*. Science 258, 1315-1319, 1992.
- Pedersen, T.M.**, (1999) *The Ramsey model of optimal economic growth*. Technical Report, Institute of Economics, University of Copenhagen, December 1999.
- Ramsey, F.P.**, (1928) *A mathematical theory of saving*. Economic Journal, vol.38, no.152, 1928, pp.543-559.